Anforderungen an Messkonzepte aus Sicht der Biostatistik

Requirements for measurement concepts from a biostatistical perspective

Hans-Peter Piepho

Fachgebiet Bioinformatik
Universität Hohenheim

Content

1. Which data?
2. Bioavailability
3. Two popular linear models / assays
4. Test for linearity
5. Optimal design in case of linearity
6. Standard errors
7. Example 1: Finney (1978)
8. Example 2: Fritz (1974)
9. Summary

Example

Collaborative Study of the Rat Hemoglobin Repletion Test for Bioavailability of Iron

JAMES C. FRITZ, GWENDOLYN W. PLA, BERTHA N. HARRISON, and GENEVA A. CLARK
Division of Nutrition, Food and Drug Administration, Washington, D.C. 20204

- 4 sources of iron
- Standard: $FeSO_4$ (sample 1)

  Test substances:

  Electrolytically reduced iron, particle sizes 7 - 10 μm (sample 2) and

  27 - 40 μm (sample 3) ; Ferric orthophosphate (sample 4)
- Response variable: concentration of hemoglobin
- 8 laboratories (!)

Example


Haemoglobin
12
10
8
6
4
0
10
20
30
40
50
Dosis
Probe
Kontrolle
Probe 1
Probe 2
Probe 3
Probe 4

1. Which data?

x = independent variable

- Concentration of trace element / micro nutrient
- Absolute uptake of trace element / micro nutrient

y = dependent variable

- Daily weight gains
- Bone ash
- Blood values

Data transformation to meet assumptions (normality, homogeneity of variance, linearity/additivity)

2. Bioavailability

x_s = amount of standard substance for achieving a certain response of y (y_0)

x_t = amount of test substance for achieving a certain response of y (y_0)

RBV = x_s / x_t

RBV = relative bioavailability value

- RBV may depend on y_0 (this depends on regression model)
- It is convenient, if RBV is independent of y_0 , but this is not guaranteed!

2. Bioavailability



2. Bioavailability

Example

$Y = \alpha[1 - \exp(-\beta x)]$

Standard:

$y_0 = \alpha[1 - \exp(-\beta_s x_s)] \quad \Leftrightarrow \quad x_s = -\log(1 - y_0/\alpha)/\beta_s$

Test substance:

$y_0 = \alpha[1 - \exp(-\beta_t x_t)] \quad \Leftrightarrow \quad x_t = -\log(1 - y_0/\alpha)/\beta_t$

$\Rightarrow$ RBV = $x_s/x_t = \beta_t/\beta_s$ $\Rightarrow$ independent of y_0 !

2. Bioavailability

Counter example

Standard:

$$y_0 = \alpha_s[1 - \exp(-\beta_s x_s)] \quad \Leftrightarrow \quad x_s = -\log(1 - y_0/\alpha_s)/\beta_s$$

Test substance:

$$y_0 = \alpha_t[1 - \exp(-\beta_t x_t)] \quad \Leftrightarrow \quad x_t = -\log(1 - y_0/\alpha_t)/\beta_t$$

$$\Rightarrow \quad \text{RBV} = x_s/x_t = \beta_t/\beta_s \times [\ \log(1 - y_0/\alpha_s) \ / \ \log(1 - y_0/\alpha_t)\ ] \ \Rightarrow \quad \text{depends on } y_0 \text{ !}$$

It all depends on the model!

3. Two popular linear models / assays

Slope-Ratio Assay

Standard:

$$y_0 = \alpha + \beta_s x_s \quad \Leftrightarrow \quad x_s = (y_0 - \alpha)/\beta_s$$

Test substance:

$$y_0 = \alpha + \beta_t x_t \quad \Leftrightarrow \quad x_t = (y_0 - \alpha)/\beta_t$$

$$\Rightarrow \quad \text{RBV} = x_s/x_t = \beta_t/\beta_s \quad \Rightarrow \quad \text{independent of } y_0 \text{ !}$$

3. Two popular linear models / assays

Slope-Ratio Assay



3. Two popular linear models / assays

Three-Point Assay



3. Two popular linear models / assays

Standard Curve Assay



3. Two popular linear models / assays

Parallel-Lines Assay

Standard:

$$Y_0 = \alpha_s + \beta \log(x_s) \quad \Leftrightarrow \quad x_s = \exp[(Y_0 - \alpha_s)/\beta]$$

Test substance:

$$Y_0 = \alpha_t + \beta \log(x_t) \quad \Leftrightarrow \quad x_t = \exp(Y_0 - \alpha_t)/\beta]$$

$$\Rightarrow \quad \text{RBV} = x_s/x_t = \exp[(\alpha_t - \alpha_s)/\beta] \quad \Rightarrow \quad \text{independent of } y_0 \text{ !}$$

But: $x = 0$ not permissible, because log(x) not defined!

3. Two popular linear models / assays

Parallel-Lines Assay



4. Test of linearity

Test of linearity

- All of these assays assume linearity
- Can test lack-of-fit of linear model
- This requires true replication at each x-level

Modelling approach:

$y_{ij} = \alpha + \beta x_i + e_{ij}$

$y_{ij} = \alpha + \beta x_i + \delta_i + e_{ij} \quad \Leftrightarrow \quad y_{ij} = \alpha + \tau_i + e_{ij}$ (one-way ANOVA)

δ_i = Effect for non-linearity (lack-of-fit)

4. Test of linearity



Abb.: Graphical display of variance decomposition for test of linearity.

5. Optimal design in case of linearity

Objective criterion:

$$\mathrm{var}\left(\hat{\beta}\right) = \frac{s^2}{SS_x}$$

s^2 = residual variance

$$SS_x = \sum_{i=1}^{n} \left(x_i - \bar{x}\right)^2$$

$\Rightarrow$ minimize SS_x

$\Rightarrow$ 50% of x-values at lower end of range, 50% at upper end of range

$\Rightarrow$ Three-Point Assay for standard and test substance !

6. Standard error

Slope-Ratio Assay: RBV = β_t/β_s

Parallel Lines Assay: RBV = $\exp[(\alpha_t - \alpha_s)/\beta]$

$\Rightarrow$ Non-linear function of parameters

$\Rightarrow$ Standard error and confidence intervals not available from basic procedures for linear models

3 methods:

(1) Fiducial Limits = Fieller's method
(2) Delta method (method of differentials)
(3) Reparametrisation & nonlinear regression

6. Standard error

Reparametrisation

Slope-Ratio Assay:

RBV = $\beta_t/\beta_s \Leftrightarrow \beta_t = \beta_s$ RBV

Parallel Lines Assay:

RBV = $\exp[(\alpha_t - \alpha_s)/\beta] \Leftrightarrow \beta = \log(\text{RBV})/(\alpha_t - \alpha_s)$

$\Rightarrow$ Package for nonlinear regression produces standard error for RBV

7. Example 1

Finney (1978, Table 7.3.1)

- Assay with nicotinic acid in meat extract
- 2 replicates (test tubes) per treatment
- Dose of test substance in ml per test tube (3 doses)

  Dose of standard in μg per test tube (5 doses)
- Cultures of *Lactobacillus arabinosus*
- Incubation at 37°C for 72 hours
- Response: Acidity, assessed by titration with NaOH

7. Example 1


10
8
6
4
2
NaOH
0.0
0.5
1.0
1.5
2.0
Dosis
Behandlung
Kontrolle
Standard
Testsubstanz

7. Example 1

```
data a;
input Dosis Behandlung$13. NaOH;
datalines;
0.05 Standard     3.5
0.05 Standard     3.2
0.10 Standard     5.0
0.10 Standard     4.7
0.15 Standard     6.2
0.15 Standard     6.1
0.20 Standard     8.0
0.20 Standard     7.7
0.25 Standard     9.4
0.25 Standard     9.5
1.0  Testsubstanz 4.9
1.0  Testsubstanz 4.8
1.5  Testsubstanz 6.3
1.5  Testsubstanz 6.5
2.0  Testsubstanz 7.7
2.0  Testsubstanz 7.7
0.0  Kontrolle    1.5
0.0  Kontrolle    1.4
;
```

7. Example 1

```
proc glm data=a;
class Behandlung;
model NaOH=Behandlung*Dosis / solution;
run;
```

Output:

Parameter	Estimate		Standard Error
Intercept	1.64533333		0.08451301
Dosis*Behandlung Kontrolle	0.00000000	B	.
Dosis*Behandlung Standard	31.08000000	B	0.56781844
Dosis*Behandlung Testsubstanz	3.09600000	B	0.06954327

RBV = $x_s/x_t = \beta_t/\beta_s$ = 3.096/31.08 = 0.0996

7. Example 1

Reparametrisation for Slope-Ratio Assay:

$\beta_t = \beta_s$ RBV

```
proc nlin data=a;
parms alpha=2 beta_s=30 RBV=0.1;
beta_t=RBV*beta_s;
if Behandlung='Kontrolle   ' then eta=alpha;
if Behandlung='Standard    ' then eta=alpha + beta_s*Dosis;
if Behandlung='Testsubstanz' then eta=alpha + beta_t*Dosis;
model NaOH=eta;
run;
```

7. Example 1

Output:

Parameter	Estimate	Approx Std Error	Approximate 95% Confidence Limits	
alpha	1.6453	0.0845	1.4652	1.8255
beta_s	31.0800	0.5678	29.8697	32.2903
RBV	**0.0996**	**0.00183**	**0.0957**	**0.1035**

For comparison „Fiducial Limits" of Finney:

(0.0957 ; 0.1026)

7. Example 1


Studentized Residuals for NaOH
Residual
Predicted Mean
Percent
Residual
Residual
Quantile
Residual Statistics
Observations 18
Minimum -1.615
Mean -0.016
Maximum 1.8584
Std Dev 1.0261
Fit Statistics
Objective -7.091
AIC -5.091
AICC -4.784
BIC -4.383

7. Example 1

Test of linearity

```
proc glm data=a;
class Behandlung Dosis_Klasse;
model NaOH=Behandlung*Dosis Behandlung*Dosis_Klasse;
run;
```

Output:

Source	DF	Type I SS	F Value	Pr > F
Dosis*Behandlung	2	96.41160000	2479.16	<.0001
Behandlung*Dosis_Klasse	6	0.27840000	2.39	0.1162

Conclusion: no evidence of non-linearity

But: Absence of evidence is not evidence of absence (Altman & Bland, 1995) !!!

7. Example 1

Finney's ANOVA

```
data b;
set a;
if Behandlung='Kontrolle' then Kontrolle_vs_Rest='Kontrolle';
else                           Kontrolle_vs_Rest='Rest     ';
run;

proc glm data=b;
class Behandlung Dosis_Klasse Kontrolle_vs_Rest;
model NaOH=Behandlung*Dosis
      Kontrolle_vs_Rest
      Behandlung
      Behandlung*Dosis_Klasse;
run;
```

7. Example 1

Source	DF	Type I SS	F Value	Pr > F
Dosis*Behandlung	2	96.41160000	2479.16	<.0001
Kontrolle_vs_Rest	1	0.14468933	7.44	0.0233
Behandlung	1	0.02487734	1.28	0.2872
Behandlung*Dosis_Klasse	4	0.10883333	1.40	0.3093

Conclusion from the 4 F-tests:

- The two slopes are significantly different from zero
- The control has a significantly different intercept than the two substances
  $\Rightarrow$ Linearity not up to dose = 0 $\Rightarrow$ exclude control from analysis!
- The two substances have the same intercept
- No evidence of non-linearity

7. Example 1

Analysis without control

Parameter	Estimate	Approx Std Error	Approximate 95% Confidence Limits	
alpha	1.8204	0.1023	1.5993	2.0414
beta_s	30.1253	0.6296	28.7652	31.4854
RBV	**0.0992**	**0.00166**	**0.0956**	**0.1028**

For comparison „Fiducial Limits" of Finney:

(0.0957 ; 0.1026)

8. Example 2

Collaborative Study of the Rat Hemoglobin Repletion Test for Bioavailability of Iron

JAMES C. FRITZ, GWENDOLYN W. PLA, BERTHA N. HARRISON, and GENEVA A. CLARK
Division of Nutrition, Food and Drug Administration, Washington, D.C. 20204

- 4 sources of iron
- Standard: $FeSO_4$ (sample 1)

  Test substances:

  Electrolytically reduced iron, particle sizes 7 - 10 μm (sample 2) and

  27 - 40 μm (sample 3) ; Ferric orthophosphate (sample 4)
- Response variable: concentration of hemoglobin
- 8 laboratories (!)

8. Example 2

Table 2. Average hemoglobin values (g/100 ml) reported by collaborators

Lab.	Neg. control	Sample 1, mg Fe/kg			Sample 2, mg Fe/kg			Sample 3, mg Fe/kg			Sample 4, mg Fe/kg		
		6	12	24	6	12	24	12	24	48	12	24	48
1	5.81	7.30	9.33	10.98	7.30	8.21	10.03	7.83	7.91	9.51	6.73	8.01	10.40
2	5.44	6.53	8.57	11.26	6.10	8.04	9.52	6.76	8.76	11.14	7.91	8.82	10.81
3	6.40	8.13	9.80	11.53	7.26	8.06	10.35	7.37	8.02	9.37	7.95	9.56	10.14
4	6.34	8.53	9.88	12.20	7.16	8.88	9.99	7.57	8.28	10.58	8.40	9.47	11.76
5	3.86	5.08	6.83	11.00	4.49	6.32	9.48	5.64	8.29	9.67	5.42	7.19	9.37
6	5.04	6.46	8.23	10.96	5.62	6.57	8.18	6.17	6.79	9.75	6.30	7.89	10.63
7	3.32	4.72	5.63	7.47	4.01	4.08	6.23	3.75	4.29	6.27	3.97	5.02	7.81
8	5.58	7.79	9.12	10.88	7.03	7.91	9.52	7.77	9.20	10.79	7.72	8.90	11.43
Av.	5.22	6.82	8.42	10.79	6.12	7.26	9.16	6.61	7.69	9.64	6.80	8.11	10.29

8. Example 2


Haemoglobin
12
10
8
6
4
0
10
20
30
40
50
Dosis
Probe
Kontrolle
Probe 1
Probe 2
Probe 3
Probe 4

8. Example 2

```
data a;
set a;
Dosis_Klasse=Dosis;
Kontrolle_vs_Rest='Rest      ';
if Probe='Kontrolle' then Kontrolle_vs_Rest='Kontrolle';
run;

proc glm data=a;
class Probe Dosis_Klasse Labor Kontrolle_vs_Rest;
model Haemoglobin=Labor Probe*Dosis Kontrolle_vs_Rest Probe
                  Probe*Dosis_Klasse;
run;
```

8. Example 2

Source	DF	Type I SS	F Value	Pr > F
Labor	7	156.0565298	72.37	<.0001
Dosis*Probe	4	268.2489657	217.69	<.0001
Kontrolle_vs_Rest	1	0.5721661	1.86	0.1766
Probe	3	0.9804307	1.06	0.3702
Probe*Dosis_Klasse	4	0.6274491	0.51	0.7291

8. Example 2


Studentized Residuals for Haemoglobin
Residual
Predicted Mean
Percent
Residual
Residual
Quantile
Residual Statistics
Observations 104
Minimum -2.36
Mean -0.002
Maximum 2.8535
Std Dev 1.0089
Fit Statistics
Objective 208.69
AIC 210.69
AICC 210.73
BIC 213.21

8. Example 2

```
data a;
set a;
array L L1-L8;
do i=1 to 8; L[i]=0; end;
L[Labor]=1;
run;

proc nlin data=a;
parms alpha_1-alpha_8=2 beta_1=30 RBV_2=0.1 RBV_3=0.1 RBV_4=0.1;
beta_2=RBV_2*beta_1;
beta_3=RBV_3*beta_1;
beta_4=RBV_4*beta_1;
alpha = L1*alpha_1 + L2*alpha_2 + L3*alpha_3 + L4*alpha_4
      + L5*alpha_5 + L6*alpha_6 + L7*alpha_7 + L8*alpha_8;
if Probe='Kontrolle' then eta=alpha;
if Probe='Probe 1  ' then eta=alpha + beta_1*Dosis;
if Probe='Probe 2  ' then eta=alpha + beta_2*Dosis;
if Probe='Probe 3  ' then eta=alpha + beta_3*Dosis;
if Probe='Probe 4  ' then eta=alpha + beta_4*Dosis;
model Haemoglobin=eta;
run;
```

8. Example 2

Parameter	Estimate	Approx Std Error	Approximate 95% Confidence Limits	
alpha_1	5.9458	0.1759	5.5965	6.2950
alpha_2	5.9696	0.1759	5.6204	6.3189
alpha_3	6.2988	0.1759	5.9496	6.6481
alpha_4	6.6911	0.1759	6.3419	7.0404
alpha_5	4.6604	0.1759	4.3111	5.0096
alpha_6	5.1181	0.1759	4.7688	5.4673
alpha_7	2.6550	0.1759	2.3057	3.0042
alpha_8	6.2758	0.1759	5.9265	6.6250
beta_1	0.2273	0.00908	0.2093	0.2454
RBV_2	0.6678	0.0385	0.5914	0.7441
RBV_3	0.3905	0.0200	0.3508	0.4302
RBV_4	0.4544	0.0211	0.4124	0.4963

8. Example 2

Modelling heterogeneity between labs

Model for *i*-th lab:

Control: $y_{ij0} = \alpha_i + e_{ij0}$

Sample 1: $y_{ij1} = \alpha_i + \beta_{1i} x_{1j} + e_{ij1}$

Sample 2: $y_{ij2} = \alpha_i + \beta_{2i} x_{2j} + e_{ij2}$

Sample 3: $y_{ij3} = \alpha_i + \beta_{3i} x_{3j} + e_{ij3}$

Sample 4: $y_{ij4} = \alpha_i + \beta_{4i} x_{4j} + e_{ij4}$

8. Example 2

Regression parameters for *i*-th lab:

$$\alpha_i = \alpha + u_{0i}$$

$$\beta_{1i} = \beta_1 + u_{1i}$$

$$\beta_{2i} = \beta_2 + u_{2i}$$

$$\beta_{3i} = \beta_3 + u_{3i}$$

$$\beta_{4i} = \beta_4 + u_{4i}$$

8. Example 2

Random regressions:

$$\begin{pmatrix} u_{0i} \\ u_{1i} \\ u_{2i} \\ u_{3i} \\ u_{4i} \end{pmatrix} \sim MVN\left[\begin{pmatrix} 0 \\ 0 \\ 0 \\ 0 \\ 0 \end{pmatrix}, \begin{pmatrix} \sigma_0^2 & \sigma_1 & \sigma_1 & \sigma_1 & \sigma_1 \\ \sigma_1 & \sigma_1^2 & \sigma_2 & \sigma_2 & \sigma_2 \\ \sigma_1 & \sigma_2 & \sigma_1^2 & \sigma_2 & \sigma_2 \\ \sigma_1 & \sigma_2 & \sigma_2 & \sigma_1^2 & \sigma_2 \\ \sigma_1 & \sigma_2 & \sigma_2 & \sigma_2 & \sigma_1^2 \end{pmatrix}\right]$$

MVN = Multivariate normal distribution

8. Example 2

```
proc nlmixed data=a maxiter=1000;
parms alpha=5.45 beta_1=0.23 RBV_2=0.7 RBV_3=0.4 RBV_4=0.4 s2e=0.33
      var0=.1 var1=.1 cov1=0 cov2=0;
beta_2=RBV_2*beta_1;
beta_3=RBV_3*beta_1;
beta_4=RBV_4*beta_1;
random u0 u1 u2 u3 u4
       ~ normal([0,0,0,0,0], [var0,
                              cov1, var1,
                              cov1, cov2, var1,
                              cov1, cov2, cov2, var1,
                              cov1, cov2, cov2, cov2, var1])
       subject=Labor;
alpha_i=alpha  + u0;
beta_1i=beta_1 + u1;
beta_2i=beta_2 + u2;
beta_3i=beta_3 + u3;
beta_4i=beta_4 + u4;
if Probe='Kontrolle' then eta=alpha_i;
if Probe='Probe 1  ' then eta=alpha_i + beta_1i*Dosis;
if Probe='Probe 2  ' then eta=alpha_i + beta_2i*Dosis;
if Probe='Probe 3  ' then eta=alpha_i + beta_3i*Dosis;
if Probe='Probe 4  ' then eta=alpha_i + beta_4i*Dosis;
model Haemoglobin ~ normal(eta, s2e);
run;
```

8. Example 2

Parameter Estimates

Parameter	Estimate	Standard Error	Lower	Upper
alpha	5.4518	0.4573	3.9964	6.9072
beta_1	0.2273	0.009913	0.1958	0.2589
RBV_2	0.6678	0.03516	0.5559	0.7796
RBV_3	0.3905	0.02621	0.3071	0.4739
RBV_4	0.4544	0.02590	0.3719	0.5368
s2e	0.1977	0.03396	0.08965	0.3058
var0	1.6192	0.8323	-1.0295	4.2680
var1	0.000358	0.000208	-0.00030	0.001020
cov1	-0.00650	0.009599	-0.03705	0.02405
cov2	0.000302	0.000202	-0.00034	0.000945

⇒ Confidence intervals wider than in Fritz et al. (1974)

9. Summary

- Functional form of dose-response curve is crucial $\Rightarrow$ model selection
- Test of linearity / lack of fit testing requires true replication

  (but: absence of evidence is not evidence of absence)
- Adequacy of model (normality, homogeneity of variance) should routinely be checked using residual plots
- Inference for bioavailability parameters is non-linear (and often non-trivial)
- Optimal design depends on shape of dose-response (choice of doses, sample size)
- All design effects (e.g., blocks, laboratories, e.t.c.) should be in the model
- Meta-analysis integrating data from several labs / studies requires random-effects modeling of heterogeneity

Literature

Finney, D.J. (1978): Statistical method in biological assay. Third Edition. Charles Griffin, London.

Littell, R.C., Lewis, A.J., Henry, P.R. (1995): Statistical evaluation of bioavailability assays. pp. 5-33. In: Ammerman, C.B. (ed.): Bioavailability of nutrients for animals: amino acids, minerals, and vitamins. Academic Press, San Diego.

Sauer, N., Emrich, K., Piepho, H.P., Lemme, A., Redshaw, M., Mosenthin, R. (2008): Meta-analysis on the relative efficiency of Methionine-Hydroxy-analogue-free-acid compared with DL-Methionine in broilers using nonlinear mixed models. *Poultry Science* **87**, 2023-2031 (Correction: **87**, 2069).